
Abstract
For a finite diagonal Abelian subgroup A ⊂ SL(4,C), assume a

crepant resolution exists; there is then an established McKay corre-
spondence on the level of Euler characteristic, Betti numbers or even
motivic integration, but the BKR result does not even begin to make
sense for an equivalence of derived categories. A necessary preliminary
to any progress on this question is a crepant resolution representing a
good functor. With this in view, I show how to compute A-HilbA4,
and establish a small catalogue of things that go wrong with it – dis-
crepancy, singularity, reducibility.

I’m mostly interested in computations, so it is convenient to stick to the
subgroup 1

r
(1, a, b, c) with 1+a+b+c = r, that is, the group µr of rth roots of

unity acting with the characters ε : (x, y, z, t) 7→ (εx, εay, εbz, εdt). The case
of a general finite diagonal Abelian subgroup A ⊂ SL(4,C) is presumably
no harder, but involves disproportionately tedious notation. (For example,
a scheme for numbering the elements and characters of A; in my restricted
case, these are 0, . . . , r − 1, with no thought involved.)

1 The condition JunSuff

As usual, the elements of the group correspond one-to-one with points of the
lattice L = Z

4 + Z · 1
r
(1, a, b, c) in the unit cube of R4 with the standard

basis (after I choose a primitive root of 1). Explicitly, the r points are
Pi = 1

r
(i, ai, bi, ci) for i = 0, . . . , r − 1 where is smallest positive residue

mod r; they divide as usual by age:

agePi =
1

r
(i+ ai+ bi+ ci) = 0, 1, 2, 3,

with the identity (or origin) at age 0, the juniors Jun at age 1, the middle-
aged Mid at age 2, and the seniors at age 3.

I make the extra assumption to ensure enough juniors to do the work:

every Pi ∈ Mid is a sum of two juniors (JunSuff)

The following is an experimental result: JunSuff is a necessary and suf-
ficient condition for the existence of a crepant resolution. More precisely,
when JunSuff holds, the barycentric subdivision at a suitably chosen junior
point leads to a fan of four cones still satisfying JunSuff; this holds for all
my 1

r
(1, a, b, c) cases with r ≤ 100.

1.1 When does JunSuff hold?

When JunSuff holds is currently slightly mysterious. Experimental evidence
(you see the pattern?): for r � 0,

JunSuff(r, [1, 2, 3, r − 6]) ⇐⇒ 6 | r
JunSuff(r, [1, 2, 4, r − 7]) ⇐⇒

r ∈ [0, 2, 4, 7, 8, 11, 14, 15, 16, 18, 22, 23] mod 28

JunSuff(r, [1, 2, 5, r − 8]) ⇐⇒ r ∈ [0, 4, 8, 10, 18, 20, 24, 28, 34] mod 40

JunSuff(r, [1, 2, 6, r − 9]) ⇐⇒ 3 | r
JunSuff(r, [1, 2, 7, r − 10]) ⇐⇒

r ∈ [0, 10, 12, 14, 20, 24, 34, 40, 42, 52, 54, 62] mod 70

JunSuff(r, [1, 2, 8, r − 11]) ⇐⇒ r ∈ [0, 4, 8, 11, 12, 15, 19, 22, 23, 24, 30, 35,

44, 46, 48, 52, 55, 56, 59, 63, 67, 68, 70, 78, 79, 88] mod 88

JunSuff(r, [1, 2, 9, r − 12]) ⇐⇒ 6 | r
JunSuff(r, [1, 3, 4, r − 8]) ⇐⇒

r ∈ [0, 8, 12, 20, 24, 32, 36, 44, 48, 56] mod 80

JunSuff(r, [1, 3, 5, r − 9]) ⇐⇒ r ∈ [0, 9, 12] mod 30

JunSuff(r, [1, 3, 6, r − 10]) ⇐⇒ r ∈ [0, 2, 3, 5, 6, 10, 11, 12,

15, 16, 20, 21, 22, 23, 26] mod 30

JunSuff(r, [1, 3, 7, r − 11]) ⇐⇒ r ∈ [0, 6, 11, 14, 18, 39, 56, 62, 69, 77, 83,

84, 95, 102, 105, 111, 116, 132, 144, 146, 161, 165,

168, 179, 182, 188, 209, 210, 216, 221] mod 231

The proper solutions with r = 20 are listed at the start of Section 3.
Those with r = 23 are:

1

23
(1, 1, 3, 18),

1

23
(1, 1, 7, 14),

1

23
(1, 2, 4, 16).

The prime (or coprime) cases, corresponding to isolated fixed points, tend
to be less rich than the composite cases. They eventually show all the same
complications as the composite cases, but hidden inside after blowups, and
so are somewhat deceptive to work with. Typical features appear at once in
composite cases, whereas in coprime cases they may only appear for large r.

The above results were obtained from the following simple Magma rou-
tines (online version for copy-and-paste at

http://www.warwick.ac.uk/~masda/McKay/Magma_AH4.)

1.2 Points of 1
r(a1, a2, a3, a4) in unit cube

function Pts(r,A)

return [[a*k mod r : a in A] : k in [1..r-1]];

end function;

1.3 The condition JunSuff

The condition is that every middle-aged point is a sum of 2 juniors. The
function returns true, or false together with the first case it fails.

function JunSuff(r,A)

Points := Pts(r,A);

Jun := [P : P in Points | &+P eq r];

Mid := [P : P in Points | &+P eq 2*r];

Sums := [[P1[i]+P2[i] : i in [1..4]] : P1 in Jun, P2 in Jun];

if &and[P in Sums : P in Mid] then return true;

else return false, [P : P in Mid | P notin Sums][1];

end if;

end function;

1.4 Search for solutions of JunSuff

For each r, list 1
r
(a1, a2, a3, a4) for which JunSuff holds

function Solutions(r);

TempSolutions := [[0,0,0,0]]; // initiate list with nonsense

for a1 in [i : i in [0..r div 4]],

a2 in [i : i in [a1..(r-a1) div 3]],

a3 in [i : i in [a2..(r-a1-a2) div 2]] do

a4 := r-a1-a2-a3;

A := [a1,a2,a3,a4];

if

(a4 ne r) and (GreatestCommonDivisor([r] cat A) eq 1) and

(&and[A ne Sort(B) : B in Pts(r,C), C in TempSolutions])

and (JunSuff(r,A)) then

Append(~TempSolutions,A);

end if;

end for;

return Remove(TempSolutions,1); // take out initial nonsense

end function;

1.5 Verifying the table

To see which 1
r
(1, 2, 7, r − 10) satisfy JunSuff, run the following routine over

the intervals [70, 139], [140, 209] and [210, 279]:

[r-70 : r in [70..139] | JunSuff(r,[1,2,7,r-10])];

1.6 JunSuff implies crepant resolution

The following routine verifies experimentally that each 1
r
(a1, . . . , a4) satisfy-

ing JunSuff has at least one barycentric subdivision at a junior point Pi that
defines a crepant partial resolution with 4 affine pieces each of which is a
cyclic quotient point (such as 1

a1
(−r, a2, a3, a4)) still satisfying JunSuff.

Jun := [P : P in Pts(r,A) | &+P eq r];

[B : B in Jun | &and [JunSuff(B[i], [-r]

cat Remove(B,i)) : i in [1..4]]];

1.7 Example: 1
23(1, 2, 4, 16)

This satisfies JunSuff, and of his 6 junior points

(1, 2, 4, 16), (2, 4, 8, 9), (3, 6, 12, 2), (6, 12, 1, 4), (12, 1, 2, 8), (13, 3, 6, 1),

barycentric subdivision at P1, P3, P6, P12 works. P13 does not work, because
1
13

(3, 3, 6, 1) has middle aged point 1
13

(9, 9, 5, 3) that is not a sum of two
juniors.

The following routine verifies all solutions for given r.

time for r in [20..25] do

&and

[&or

[&and [JunSuff(B[i], [-r] cat Remove(B,i)) : i in [1..4]]

: B in Jun ]

: A in Solutions(r)];

end for;

This takes approx 2 seconds to do [20..25], but approx 40 seconds to do the
80 cases with r = 56.

2 A-sets and monomial A-clusters

For a finite diagonal subgroup A ⊂ GL(n), the ring of invariants (whose Spec
is the quotient X = An/A) is based by invariant monomials, that make up the
sublattice of monomials M ⊂ Zn dual to the overlattice L = Zn +

∑
Z · 1

r
(a)

(summed over 1
r
(a) ∈ A). Each character i ∈ Â = Hom(A,Gm) gives rise

to the eigensheaf Li, based by monomials forming a coset of M in Zn. This
section turns these conceptually simple ideas into practical calculations.

2.1 Invariant monomials

It is easy to say what to do: put in all the invariant monomials in the unit
cube except those that are divisible in a nontrivial way by other invariant
monomials. However, I streamline a bit to avoid carrying out O(r3) tests on
O(r4) monomials. I have no doubt that this could all be done more cleverly,
for example, by using continued fractions to find monomials in two variables
in a given eigenspace.

When testing for divisibility, it is efficient to list the invariant monomials
by degree. The routines below assume that A ⊂ SL(4) and A[1] = 1.

KK := FiniteField(1009);

RR<x,y,z,t> := PolynomialRing(KK,4);

function InvariantMonomials(r,A)

Inv := [x^r,y^r,z^r,t^r,x*y*z*t];

for i in [0..r-1], j in [0..r-i-1], k in [0..r-i-j-1] do

if i*j*k eq 0 then

Append(~Inv,x^((-A[2]*i-A[3]*j-A[4]*k) mod r)*y^i*z^j*t^k);

elif (A[2]*i+A[3]*j+A[4]*k) mod r eq 0 then

Append(~Inv,y^i*z^j*t^k);

end if;

end for;

Exclude(~Inv,1); // Omit 1 before testing divisibility.

S := [[RR!1]]; // S is a list of lists Si of mons of deg i.

for i in [1..r] do

Si := [m : m in Inv | TotalDegree(m) eq i];

Append(~S,Si); // Omit those divisible by mons of deg i.

Inv := [m : m in Inv |

not &or[IsDivisibleBy(m,n) : n in Si]];

end for;

if #Inv ne 0 then

error("Invariant monomial has degree > r");

end if;

return S;

end function;

2.2 Little open question

I assume tacitly that the semigroup of monomials is generated by elements
xm = xm1ym2zm3tm4 of total degree

∑
mi ≤ r. This holds in two variables

by Jung–Hirzebruch continued fractions games, and for three variables and
A ⊂ SL(3), but in my context I take this as an experimental fact (and test
for it at the end of InvariantMonomials in case it might fail).

2.3 Eigenmonomials

Next, I put in the monomials that generate the eigensheaves Li; I test for
divisibility by banging monomials into the quotient ring (or coinvariant ring)
R = R/{invariant monomials}.

function EigenSp(r,A)

Inv := InvariantMonomials(r,A);

Eig := [&cat[S : S in Inv]]; // Collate the list of lists.

Rbar := quo<RR | Exclude(Eig[1],1)>; // invariants <> 1

for s in [1..r-1] do

temp := [x^((s-A[2]*i-A[3]*j-A[4]*k) mod r)*y^i*z^j*t^k :

k in [0..r-i-j-1], j in [0..r-i-1], i in [0..r-1] |

i*j*k eq 0] cat [y^i*z^j*t^k :

k in [0..r-i-j-1], j in [0..r-i-1], i in [0..r-1] |

(i*j*k ne 0) and ((A[2]*i+A[3]*j+A[4]*k) mod r eq s)];

temp := [m : m in temp | Rbar!m ne 0];

Append(~Eig,temp);

end for;

return Rotate(Eig,-1); // Put invariants in final rth place.

end function;

2.4 Monomial A-clusters

An A-cluster Z has H0(OZ) = k[A], the regular representation of A. There-
fore H0(OZ) has a monomial basis with exactly one monomial in each eigen-
space. In other words, to make Z, pick one monomial in each eigenspace and
set each of its compatriots to be a multiple of it, by an A-invariant rational
function that is a candidate for a function on an affine piece of A-Hilb. If I
can do that at all, I can certainly set all these multiples to zero, obtaining the
monomial cluster that is a zero dimensional strata of the toric scheme A-Hilb.
The choice of monomial basis is called an A-set (Nakamura’s A-graph).

Listing all possible A-sets is again conceptually simple: mark one mono-
mial in each eigenspace, let all the unmarked guys (in all eigenspaces) gen-
erate an ideal, and ask whether this generates an A-cluster. This fails if
my choice kills off a whole eigenspace; for example, if A = 1

5
(1, 4), choosing

y4 ∈ Eig(1) kills off x, so excludes the choice of x2 ∈ Eig(2). Or, as an even
more extreme case, choosing y4 ∈ Eig(1) and x4 ∈ Eig(4) would kill off both
x and y.

2.5 Tree search

The computational difficulty: a search over the Cartesian product of r eigen-
spaces, each containing O(r2) elements, is not practical. Instead, my routine
below is a tree search: starting from a single choice of monomial m1 ∈ Eig(1),

make a number of appropriate choices of monomials mij ∈ Eig(ij) for j =
1, . . . , ν. This is a typical node of my tree. From it, the algorithm construct
the quotient ring R by killing off all the compatriots of my choices:

R = R/({Eig(ij) \mij

∣∣ j = 1, . . . , ν}).

For a subsequent value of i (starting from iν + 1), consider the number:

ni = #{m ∈ Eig(i)
∣∣ 0 6= m ∈ R}.

The case ni = 0 is a dead end. There is no monomial in Eig(i) with nonzero
residue in R; my choices so far are therefore contradictory (as described above
for 1

5
(1, 4)), and I backtrack to the last branching point whose possibilities are

not exhausted. The case ni = 1 is a straight track: Eig(i) contains a single
appropriate monomial: there is no contradiction and no choice to make, so
simply pass on to the next i. The case ni ≥ 2 is a branching point: I list
the choices, and restart the search from this node (making the first choice

from an ordered list). If I get to the end of the list Â of eigenvalues (in my
A = Z/r case, i = r − 1), then each eigenspace of R is one dimensional; this
means that my choices so far is an A-set, so pick the fruit (thus giving the
node the logical status of dead end) and backtrack.

2.6 Preliminaries to ASets

Given a quotient ring R and a finite set of monomials S, list m ∈ S that
map to nonzero elements of R.

function NonZeroMonomials(Rbar,S)

return [m : m in S | Rbar!m ne 0];

end function;

Given a quotient ring R and a list Eig of finite sets of monomials, lists
the i for which Eig[i] has 0 or ≥ 2 monomials.

function NotOne(R,E)

return [[i] cat L : i in [1..#E] | #L ne 1 where

L is NonZeroMonomials(R,E[i])];

end function;

2.7 The main ASets function

function ASets(r,A)

Eig := EigenSp(r,A);

// initialise dynamic constructions

Bag := []; // To store the fruit.

finished := false;

I := []; // List of indexes i1, i2, ... where I make choices.

C := []; // List of lists of candidates mi in Eig[i], i in I.

// Cycle through monomials in last list then delete it.

M := []; // List of number of current choices from C.

// In other words, current choices are C[i,M[i]] for i in I.

while not finished do

// Make the current quotient ring by killing all monomials

// in all Eig[i] except the current choices mi for i in I.

Rbar := quo< RR | Exclude(Eig[r],1) cat

&cat[Exclude(Eig[I[i]],C[i,M[i]]) : i in [1..#I]] >;

// Analyse whether it is an A-cluster.

tlist := NotOne(Rbar,Eig);

if tlist eq [] then

Include(~Bag,Basis(DivisorIdeal(Rbar))); end if;

// If so, bag the fruit (intending to backtrack).

// If each tlist[i] is [i, plus >= 1 monomial], so no

// contradiction or result, go forward: add new index to I,

// new list of candidates to C, and first choice to M.

if (tlist ne []) and (&and[#l ne 1 : l in tlist]) then

Append(~I,Integers()!tlist[1,1]);

Append(~C,Remove(tlist[1],1)); Append(~M,1);

end if;

// I,M; // Uncomment this line for a diagnostic display.

if (tlist eq [])

or ((tlist ne []) and (&or[#l eq 1 : l in tlist])) then

// Backtrack, include setting finished to true if can’t.

ss := #I;

if &and[M[i] eq #(C[i]) : i in [1..ss]] then

finished := true;

else tt := Max([i : i in [1..ss] | M[i] ne #(C[i])]);

I:=I[1..tt]; M:=M[1..tt]; M[tt]:=M[tt]+1; C:=C[1..tt];

end if;

end if;

end while;

return Bag;

end function;

For example, ASets(20,[1,4,5,10]) gives 27 sets of monomials ideals of
A-clusters, of which one specimen is

[x2, xy3, xyz, xt, y4, y3z2, y2t, z3, z2t, t2].

3 Deforming A-clusters, analysing the results

3.1 Main aim

My function ASets lists A-sets, each named by the monomials S generating
the ideal of a cluster ZS, so that H0(OZ) = RS = k[x, y, z, t]/S is the reg-
ular representation of A. An A-set S defines an affine toric piece of A-Hilb
parametrising A-cluster having the same eigenbasis as the monomial cluster
RS = k[x, y, z, t]/S; these are flat deformations of RS, but not necessarily
small deformations in cases when the open piece is reducible. I now show
how to treat questions such as how bad A-Hilb is (discrepant, singular, re-
ducible, and so on) in terms of algorithms starting from S. At some points I
am guided by numerological hints as to what A-Hilb should look like coming
from the Mackay correspondence in motivic integration.

3.2 Eigenbasis of RS

The quotient is a direct sum

H0(OZ) = RS =
⊕
i∈Â

k · ni

of 1-dimensional eigenspaces, each based by a monomial ni. Another way of
saying this (following Nakamura) is that the monomials under the Newton
polygon of S form a fundamental domain in Zn for translation by the lattice
M of invariants.

The function EigenBasis returns a monomial basis of RS sorted into
eigenspaces. It assumes S is a finite set of mononials defining a finite di-
mensional quotient R/S (for example, an A-set), and the polynomial ring is
divided up into a list E = Eig of eigenspaces (as produced by my function
EigenSp(r,A)).

function EigenBasis(S,E)

Rbar := quo<RR | S>;

MonB := MonomialBasis(Rbar);

return [[n : n in MonB | n in E[i]][1] : i in [1..#E]];

end function;

3.3 Matrix of relations

An A-set S generates the monomial ideal of RS; the affine piece YS of A-Hilb
replaces each monomial relation m = 0 for m ∈ S by a nearby binomial
relation m = λmn, where n is the element of the eigenbasis in the same
character as m, and λm is some scalar.

The birational view of this is as a set of invariant rational monomials
λm = m/n ∈ M . (Birational methods are not always in order, since A-Hilb
may be reducible.) These monomials generate a cone in M , whose Spec is the
corresponding affine piece YS of A-Hilb. In cases where the cone 〈λm〉 ⊂MR

contains a whole line, there is a relation M ·(
∏
λνmm −1) with M a monomial,

so the monomial ideal (with all λm = 0) is not in the closure of the birational
component, and A-Hilb is reducible.

The birational case is standard toric geometry: given S, for each m ∈ S,
calculate its character (eigenvalue) and look up its mate n in the eigenbasis
of R/S, giving rise to the set of invariant monomials λm = m/n. Assume
1
r
(A) is given and that Eig = EigenSp(r,A) is already in place.

function RelMatrix(rr,AA,SS,Eig)

Rbar := quo<RR|SS>;

EB := EigenBasis(SS,Eig);

M := Matrix(4,[Integers()|1,1,1,1]); // add invariant x*y*z*t

for m in SS do

if m eq x*y*z*t then; // don’t add invariant x*y*z*t twice

else

B := [Degree(m,RR.i) : i in [1..4]];

c := &+[AA[i]*B[i] : i in [1..4]] mod rr;

if c eq 0 then c := rr; end if;

n := EB[c];

C := [Degree(n,RR.i) : i in [1..4]];

M := VerticalJoin(M,Matrix(4,[B[i]-C[i] : i in [1..4]]));

end if;

end for;

return M;

end function;

3.4 Example: The coprime case A = 1
23(1, 2, 4, 16)

This case illustrates how to calculate A-Hilb and one of the mechanisms
that can make it nonsingular but discrepant, a simple blowup of a crepant
resolution.

The conclusion in this case is that A-Hilb is nonsingular, a toric union of
26 copies of A4, and has discrepancy an irreducible divisor E9 corresponding
to P9 = 1

23
(9, 18, 13, 6); this divisor appears on 6 affine pieces. Omitting P9

from the fan of A-Hilb gives a toric blowdown to a crepant resolution Y ,
contracting the discrepancy divisor to a toric surface P2 ⊂ Y with normal
bundle O(−1)⊕O(−2). In this case, the mechanism causing A-Hilb to blow
up P9 also has a reasonable interpretation (see the end of 3.6).

To start, use AH23 := ASets(23,[1,2,4,16]) to list the A-sets. There
are 26 of them, with 4 to 9 generators in each set. For example, the biggest
is No. 11

AH23[11] = [x2, xyzt, xyt2, xz2, y2, yz2, z4, z2t, t3].

These define a monomial ideal, and I want its neighbours in A-Hilb, generated
by the corresponding binomials

x2 − ay, xyzt− π, xyt2 − γz, etc.

Apart from my naming scheme for scalars (see below), this is given by

RelMatrix(r,A,AH23[11],Eig) =

1 1 1 1
2 −1 0 0
1 1 −3 2
1 0 2 −2

0 2 −1 0
−1 1 2 −2

0 0 4 −1
−1 0 2 1

0 −1 0 3

The nonsingularity result I want corresponds to the fact that the 9 rows of
this matrix are all positive linear combinations of rows 2, 3, 6, 9, which is a
basis of M dual to the basis

P13 =
1

23
(13, 3, 6, 1), P6 =

1

23
(6, 12, 1, 4),

P9 =
1

23
(9, 18, 13, 6), P2 =

1

23
(2, 4, 8, 9).

A simple-minded but effective way of discovering this is by linear algebra: do

> M := RelMatrix(r,A,AH23[11],Eig);

> KernelMatrix(M);

[ 1 0 0 0 -1 0 -1 1 -1]

[ 0 1 0 0 0 0 -1 2 -1]

[ 0 0 1 0 -1 0 0 1 -1]

[ 0 0 0 1 0 0 -1 1 0]

[ 0 0 0 0 0 1 0 -1 1]

This consists of the linear relations between the rows of M . For example, the
−1 in n47 and n58 express rows 7, 8 as sums of the other rows. The function
IsNonSingularPiece below does this automatically (when it works).

3.5 Checking nonsingularity

For any monomial ideal, the rows of its RelMatrix are a set of monomials in
the toric lattice M . The preferred case is when they span a basic monomial
cone; this means that 4 of the monomials are a basis of M , and the others
are positive integral combinations of them. The following Magma function
(while somewhat kludgy) usually works for this.

Given a matrix N , ask for the index j of a positive entry nij in a row
with all the other entries nij′ < 0, or vice versa. Return 0 if none.

function convex_row(N)

n := NumberOfRows(N);

J := 0;

for i in [1..n] do

if J eq 0 then

pos := [j : j in [1..n+4] | N[i,j] gt 0];

neg := [j : j in [1..n+4] | N[i,j] lt 0];

if (#neg eq 1) and (#pos gt 1) then

found := true; J := neg[1];

elif (#pos eq 1) and (#neg gt 1) then

found := true; J := pos[1];

end if;

end if;

end for;

return J;

end function;

The nonsingular case is when 4 rows of M = RelMatrix base the cone
generated by all rows. In other words, all but 4 rows of M are eliminated
as convex integral combinations of the basis. In this case M0 is formed by
these rows, standing for monomials, and plus-or-minus Adjoint(M0) is the
dual basis of L, part of the fan of A-Hilb.

function IsNonSingularPiece(r,A,Si,Eig)

M := RelMatrix(r,A,Si,Eig);

M0 := M;

good := true;

while good and (NumberOfRows(M0) gt 4) do

N := KernelMatrix(M0);

j := convex_row(N);

if j ne 0 then RemoveRow(~M0,j);

else good := false;

end if;

end while;

if good then

N0 := Transpose(Adjoint(M0)); // dual cone

if &and[N0[i,j] le 0 : j in [1..4], i in [1..4]]

then N0 := -1*N0;

end if;

Discr := [1/r*&+[N0[i,j] : j in [1..4]]-1 : i in [1..4]];

// return true, M0, N0, Discr;

return true, N0, Discr;

else

return false, M0, KernelMatrix(M0);

end if;

end function;

The above routine seems to give the right answer in all the nonsingular
cases, i.e., for basic cones of monomials. In the singular cases, it seems to
detect that the cone is not basic, but does not find all the convex dependence
relations, so the output is sometimes longer than necessary. The general toric
functions in Magma will be available in the next release, and there is no point
in improving the current functions.

3.6 Application to 1
23(1, 2, 4, 16)

The six discrepant affine pieces of A-Hilb come from

for i in [4,5,9,10,11,12] do

IsNonSingularPiece(r,A,AH23[i],Eig);

end for;

The output leads to the following neat conclusion: consider the lattice points

e2 = (0, 1, 0, 0), P2 =
1

23
(2, 4, 8, 9), P3 =

1

23
(3, 6, 12, 2),

P13 =
1

23
(13, 3, 6, 1), P6 =

1

23
(6, 12, 1, 4), P9 =

1

23
(9, 18, 3, 6).

All but the last are junior, whereas P9 has age 2. Observe that

P9 = P3 + P6 and e2 + P2 + P13 = P3 + 2P6.

That is, in the junior affine 3-space, the line P3P6 passes through the centroid
of triangle ∆e2P2P13 (draw P3 as lying over ∆ and P6 under).

The three simplexes

P3P6e2P2, P3P6P2P13, P3P6P13e2

are all basic, and provide the fan of a crepant resolution. Instead, the fan of
A-Hilb (given by the above output) has six discrepant simplexes, obtained
by slicing these simplexes along the planes

P9e2P2, P9P2P13, P9P13e2,

from P9 on the central axis P3P6. Interestingly, A-Hilb is unable to accept
the crepant solution of joining P3P6 in a fan because it is obliged to mark
the boundaries between (z3 vs x3t2), (z3 vs xyt2) and (z3 vs xt5) in Eig[12].
In each case, P3 prefers z3, and P6 prefers the other, whereas P9 is impartial
between them.

15



3.7 Applications to 12 cases with r = 20

There are 12 cases with r = 20 within the confines of my current 1
r
(1, a, b, c)

computational impediment:

> Sol20 := [A : A in Solutions(20) | A[1] eq 1];

> ASets20 := [ASets(20,B) : B in Sol20];

> [#Si : Si in ASets20];

[ 20, 24, 20, 20, 20, 20, 20, 42, 27, 33, 20, 30 ]

for i in [1..#Sol20] do

A := Sol20[i]; Eig := EigenSp(20,A);

i, #ASets20[i],

&and[IsNonSingularPiece(20,A,ASets20[i][j],Eig)

: j in [1..#ASets20[i]]];

end for;

The first line gives the twelve cases Sol20 as

(1, 1, 2, 16) (1, 1, 3, 15) (1, 1, 4, 14) (1, 1, 6, 12)
(1, 1, 8, 10) (1, 1, 9, 9) (1, 2, 2, 15) (1, 2, 5, 12)
(1, 2, 7, 10) (1, 3, 4, 12) (1, 4, 4, 11) (1, 4, 5, 10)

(Sol20)

(the only other essential case with r = 20 is 1
20

(2, 5, 5, 8)).
The second line calculates the A-sets in these twelve cases, and reports

back the number of A-sets in each case as a coarse diagnostic of how com-
plicated the A-Hilb is. I explain how a seasoned experimentalist guess-reads
this output. For 7 of these groups A there are 20 A-sets; in these cases A-Hilb
should be a crepant resolution: each A-set defines a monomial cluster that
is the origin of an affine toric piece, A4 in all the cases I have solved. The
overlap between affine pieces is a toric variety with a Gm = C

× factor, so
A-Hilb has Euler number 20, as predicted by the McKay correspondence for
a crepant resolution.

In the cases with 24, 27 and 30, I expect A-Hilb to be a fairly simple
blowup of a crepant resolution, as in 3.6. For example, blowing up a point
in a 4-fold crepant resolution increases the number of affine pieces by 3 (and
with it the Euler number); blowing up a copy of P1 increases it by 4; blowing
up a nonsingular toric surface increases it by the Euler number of the surface.
The A-Hilb itself has a good chance of being nonsingular, or having at worst
xy = ztu or (xz = yt)× A1 as singularities.

The case 1
20

(1, 2, 5, 12) with 42 affine pieces stands out, and at first sight I
expected it to be reducible (by analogy with the exuberant case 1

30
(1, 6, 10, 13)

treated below); for example, a component isomorphic (as toric variety) to
P

1×P1×P1×P1 would contribute 16 toric affine pieces. In this case A-Hilb
turns out to be irreducible, in fact with only two affine pieces having the
mildest possible singularity 3-fold node× A1.

The third line reports back that of the twelve cases, Nos. 8, 10 and 12 are
singular. Taking apart the logical conjunction &and one can say which affine
pieces are singular and how badly.

I now test these guesses by computations in particular cases. The fol-
lowing case, while too simple to be of genuine interest, illustrates most of
the terms; not much else happens in the 7 good cases with r = 20 and 20
affine pieces. The example also illustrates the way in which the A-set itself
determines an affine toric piece of A-Hilb.

3.8 Example

For 1
20

(1, 1, 2, 16), one affine piece of A-Hilb is A4 with coordinates a, b, c, d;
it parametrises the clusters

x = ay, y2 = bz, z8 = ct, t2 = dz6, z2t = e, with e = cd. (1)

The right-hand side of the equations (x, y2, z8, t2, z2t) names the A-set. The
quotient by these has monomial basis

1 z z2 . . . z7 E
t zt E
E

y yz z2 . . . yz7

yt yzt

(2)

(where E denotes an equation) as has RS = H0(OZ) for every Z in this
affine piece. The right-hand side of (1) are obtained by looking up the unique
monomial in (2) in the same eigenspace as the left-hand side; these relations
are necessarily present, because the two sides map down to a 1-dimensional
vector space. For the same reason, there must be a fifth relation z2t = r5 for
some scalar r5; I prove r5 = cd. For this, multiply z2t = r5 by t:

r5t = z2t2 = dz8 = cdt;

moreover, t is a basis of an eigenspace, so is certainly not zero.

17



Finally, one checks that xyzt = abcd modulo (1); in fact

xyzt = ay2zt = abz2t = abcd.

This is the characteristic property for an affine piece of A-Hilb to be crepant
over X = A

4/A: x, y, z, t are coordinates on A4, and a, b, c, d are invariant
rational monomials in them that form local coordinates on A-Hilb and satisfy
dx ∧ dy ∧ dz ∧ dt = da ∧ db ∧ dc ∧ dd.

3.9 Remark: socle and reducible A-Hilb

Notice that the above deduction r5 = cd is based on the fact that the right-
hand side of z2t = r5 is r5 · 1, where r5 is a scalar to be determined, and
1 ∈ R can be multiplied by some monomial t to give a basic element of
R – this argument would definitely fails if the monomial on the right-hand
side were in the socle of R, when no deduction is possible about the value
of the scalar. This should lead to exceptional components of A-Hilb, and
certainly does so in sufficiently complicated cases; these components do not
belong to the birational component of A-Hilb, and lie over the origin of C4/A
(or a small toric stratum). In other words, they parametrise A-clusters not
obtained as limits of free orbits. For an example where this really happens,
see 5.4.

4 Appendix: the case 1
20(1, 2, 5, 12)

This has 42 A-sets, each giving an affine piece Yi; just two of these are
singular, Y15 and Y17 (see below). I calculate Y8 as a typical example – it
is nonsingular but discrepant, with a close relation between the discrepancy
and the shape of the clusters parametrised by Y8. Its origin is the monomial
A-cluster defined by

(x4, x3z, x3t, x2z2, xz3, xt2, y, z4, z2t, zt3, t5) (3)

The quotient ring has monomial basis

1 t t2 t3 t4

z zt zt2

z2

z3

x xt
xz xzt
xz2

x2 x2t
x2z x2zt

x3

18



RelMatrix gives the neighbouring clusters; its output (massaged a bit) is the
matrix with 12 rows

M =

1 0 −1 2
−1 0 1 3

0 0 0 5
1 0 3 −3
4 0 0 −2
2 0 2 −1

0 0 4 0
1 1 1 1
−2 0 2 1

3 0 −3 1
3 0 1 −4
−2 1 0 0

The last four rows give the invariant monomials

κ = z2t/x2, m = x3t/z3, l = x3z/t4, b = y/x2.

The kernel matrix (linear dependencies between rows of M)

1 0 0 0 0 0 0 0 −1 −1 0 0
0 1 0 0 0 0 0 0 −2 −1 0 0
0 0 1 0 0 0 0 0 −3 −2 0 0
0 0 0 1 0 0 0 0 −1 0 −1 0
0 0 0 0 1 0 0 0 −1 −1 −1 0
0 0 0 0 0 1 0 0 −2 −1 −1 0
0 0 0 0 0 0 1 0 −3 −1 −1 0
0 0 0 0 0 0 0 1 −3 −2 −1 −1

expresses the first eight rows of M as their linear combinations, or in other
words, expresses

xt2/z, zt3/x, t5, xz3/t3, x4/t2, x2z2/t, z4, xyzt,

as monomials in κ,m, l, b (for the proof, see below). For example,

xt2/z = κm and xyzt = κ3m2lb.

One shows eventually that Y8 is A4 with coordinates κ,m, l, b. In other
words, κ,m, l, b base the cone of invariant monomials that are positive on
the monomial ideal (3). One calculates the dual basis of L:

P11 + e2 = 1
20

(11, 22, 15, 12), P9 = 1
20

(9, 18, 5, 8),

P5 = 1
20

(5, 10, 5, 0), e2 = (0, 1, 0, 0),

4.1 Observation: discrepancy and “all the equations”

The above results for Y8 are possibly typical of what holds whenever an affine
piece is nonsingular. The basis of the dual cone in L has ages 3, 2, 1, 1, which
coincides with the exponents in

xyzt = κ3m2lb,

giving the Jacobian determinant and the discrepancy as κ2m = xyzt/κmlb.
The defining ideal of the A-cluster of Y8 has minimal generators

z2t = κx2, x3t = mz3, x3z = lt4, y = bx2,

xt2 = κmz, zt3 = κ2mx, t5 = κ3m2 · 1, xz3 = κlt3,

x4 = κmlt2, x2z2 = κ2mlt, z4 = κ3ml · 1, xyzt = κ3m2lb,

(The first four are basic, and the remainder are deduced from them by can-
celling a basic monomial. See below for the proof.) However, this minimal
basis is part of a whole set of equations indexed by the 47 = 4 · 3 · 2 · 2 − 1
non-unit factors of κ3m2lb, with redundant relations such as

z4t2 = κ2x4, z6t3 = κ3x6, z3t4 = κ3mx3, etc.

Notice that in this context, each relation has a complementary relation, ob-
tained by multiplying the left-hand side by κ3m2lb and the right-hand side
by xyzt and cancelling, so xm+1 = ayn1zn2tn3 gives yn1+1zn2+1tn2+1 = αxm

with aα = π = κ3m2lb, etc. The minimal generators have complements

x3y = κ2l2mbz, yz4 = κ3l2bx2, yt5 = κ3lmbx2, x3zt = κ3l2m,

yz2 = κ2l2bt, x2y = κl2bt2, xyz = l2bt4, yt3 = κ2lmbz2,

yzt3 = κ2lbx3, yt2 = kbxz, xyt = bz3, xyzt = κ3m2lb

which are already in the ideal.
Recall Nakamura’s result that in the A ⊂ SL(3) case, every A cluster can

be defined by exactly 7 relations, including xyz = π and two complementary
sets of three relations of the form

xm+1 = ayn1 z
n
2 , yn1+1zn2+1 = αxm with aα = π, etc.

Proof of nonsingularity The idea is as in 3.8: I am allowed to cancel
basic monomials. For the relation represented by the first row of M0, and
the first row of KernelMatrix(M0): I know that a relation xt2 = r1z holds.
Since x2z is basic in the monomial cluster,

r1x
2z = x3t2 = m ∗ z3t = κmx2z,

and cancelling x2z gives r1 = κm. The others are similar.

x*z basic, so r2 *x*z = z^2*t^3 = ka*x^2*t^2

x^2 basic, x^2*t^5 = ka^2*m^2*z^2*t

t^4 basic etc.

This ideal situation may hold whenever the toric cone is basic.

4.2 Singular cases

Of the 42 affine pieces of A-Hilb for 1
20

(1, 2, 5, 12), only Y15 and Y17 are sin-
gular. For Y15, permuting the 13 rows of RelMatrix gives

M0 =

−1 0 1 3
0 2 0 −2
1 0 3 −3
0 1 2 −1
0 0 4 0
1 0 −1 2

0 0 0 5
1 1 1 1
2 −1 0 0
−1 1 −1 2

0 −1 2 1
1 1 1 −4
1 1 −3 1

The last five rows of M give the invariant monomials

a = x2/y, λ = yt2/xz, κ = z2t/y, δ = xyz/t4, γ = xyt/z3.

The kernel matrix of M (linear dependencies between its rows):

1 0 0 0 0 0 0 0 0 −1 −1 0 0
0 1 0 0 0 0 0 0 0 −1 0 −1 0
0 0 1 0 0 0 0 0 0 0 −1 −1 0
0 0 0 1 0 0 0 0 0 −1 −1 −1 0
0 0 0 0 1 0 0 0 0 −1 −2 −1 0
0 0 0 0 0 1 0 0 0 0 −1 0 −1
0 0 0 0 0 0 1 0 0 −1 −2 0 −1
0 0 0 0 0 0 0 1 0 −1 −2 −1 −1
0 0 0 0 0 0 0 0 1 1 −1 0 −1

The singularity shows up in the 1, 1,−1,−1 in the last row, which says that
aλ = κγ (initially under birational assumptions, proved later). Note that
while this equation gives Y15 the singularity (3-fold node) × A1, the other
equations are shielded from the effect of a: the ratios appearing in them are
monomials in λ, κ, δ, γ.

x2 = ay, yt2 = λxz, z2t = κy, xyz = δt4, xyt = γz3

zt3 = λκx, y2 = λδt2, xz3 = κδt3, yz2 = λκδt

z4 = λκ2δ, xt2 = κγz, t5 = λκ2γ, xyzt = λκ2δγ.

with aλ = κγ. The relation xyzt = λκ2δγ = aλ2κδ is redundant as ideal
generator.

To prove aλ = κγ, note that yz is basic, and

κγyz = γz3t = xyt2 = λx2z = aλyz.

Cases 17

========

> M0;

[-1 1 -1 2 ]

[ 0 -1 2 1 ]

[ 0 1 2 -1 ]

[-1 0 1 3 ]

[ 0 0 4 0 ]

[ 1 0 -1 2 ]

[ 1 1 1 1 ]

[ 2 -1 0 0 ]

[ 0 2 0 -2 ]

[-1 -1 -1 4 ]

[ 1 0 3 -3 ]

[ 1 1 -3 1 ]

> KernelMatrix(M0);

[ 1 0 0 0 0 0 0 0 -1 -1 0 0 ]

[ 0 1 0 0 0 0 0 0 0 -1 -1 0 ]

[ 0 0 1 0 0 0 0 0 -1 -1 -1 0 ]

[ 0 0 0 1 0 0 0 0 -1 -2 -1 0 ]

[ 0 0 0 0 1 0 0 0 -1 -2 -2 0 ]

[ 0 0 0 0 0 1 0 0 0 -1 -1 -1 ]

[ 0 0 0 0 0 0 1 0 -1 -2 -2 -1 ]

[ 0 0 0 0 0 0 0 1 1 0 -1 -1 ]

As above, read the last 4 lines of RelMatrix as

x^2 = a*y

y^2 = b*t^2

t^4 = d*x*y*z

x*z^3 = l*t^3

x*y*t = ga*z^3

with a*b = l*ga (initially under birational assumptions).

Note that a is not explicitly involved in the other equations.

y*t^2 = b*d*x*z

z^2*t = d*l*y (Pf. r*x*y*z = x*z^3*t = l*t^4 = d*l*x*y*z)

y*z^2 = b*d*l*t

z*t^3 = b*d^2*l*x

z^4 = b*d^2*l^2

x*t^2 = d*l*ga*z

x*y*z*t = b*d^2*l^2*ga (not a defining relation)

Proof that a*b = l*ga:

l*ga*t^3 = g*x*z^3 = x^2*y*t = a*y^2*t = a*b*t^3

How are Y15 and Y17 glued together?

Y15 is

a = x^2/y, ga = x*y*t/z^3

de = x*y*z/t^4

la = y*t^2/x*z

ka = z^2*t/y

with a*la = ka*ga

Y17 is

a = x^2/y, ga = x*y*t/z^3

d = t^4/x*y*z

b = y^2/t^2

l = x*z^3/t^3

with a*b = l*ga

Obviously a, ga coincide. Also d = de^-1 and b = de*la,

l = de*ka. And conversely, la = b*d, ka = l*d.

5 Counter-examples

33 cases of 1
30

(1, a, b, c) satisfy JunSuff. The number of affine pieces of their
A-Hilb is given by

> Sol30 := [A : A in Solutions(30) | A[1] eq 1];

> ASets30 := [ASets(30,B) : B in Sol30];

> [#S : S in ASets30];

[ 30, 30, 38, 30, 30, 34, 30, 30, 30, 30, 30, 30, 45, 36,

37, 34, 37, 38, 73, 38, 70, 38, 34, 63, 46, 58, 56, 50,

38, 59, 158, 56, 42 ]

Notice that 10 of these have 30 A-sets, and I expect that for these, A-Hilb is
a crepant resolution. A further 15 have ≤ 50 A-sets, and 7 have between 56
and 73, so are likely to be somewhat discrepant or singular or both.1

5.1 Reducible example

The exuberant 31st case A = 1
30

(1, 6, 10, 13) stands out clearly with its 158 A-
sets; it turns out to be reducible. Its more interesting affine pieces correspond
to those with most equations:

> r:=30; A := Sol30[31]; A // Answer: [ 1, 6, 10, 13 ]

> AS := ASets(r,A); [i : i in [1..158] | #AS[i] ge 17];

This says that A-sets numbered 25, 29, 30, 31, 32, 33, 51, 52, 84, 89, 90, 99, 107
have ≥ 17 defining equations. AS[25] gives

x6, x5t, x4t2, x3y, x2z, xy2, xyt, xzt, xt3, y5, y3t, yz, yt2, z3, z2t, zt2, t5

1I have computed lots of these cases, and am beginning to understand some of the
mechanisms.

and MonomialBasis(quo<RR|AS[25]>) gives:

1 x x2 x3 x4 x5

t xt x2t x3t x4t
t2 xt2 x2t2 x3t2

t3

t4

y xy x2y
yt

y2

y2t
y3

y4

z xz z2 xz2

zt

(4)

The function RelMatrix gives the monomial relations; I always include
the relation for xyzt, although it is redundant in many cases, because it
computes the discrepancy.

1 1 1 1
6 −1 0 0
5 −3 0 1
4 0 0 2
3 1 0 −3
2 −2 1 0
1 2 0 −1
1 1 −2 1
1 −4 1 1

1 0 −1 3
0 5 0 0
−1 3 0 1
−3 1 1 −1
−2 1 0 2

0 0 3 0
−3 0 2 1

0 −1 1 2
−5 0 0 5

It turns out that all the functions on this affine piece of A-Hilb are gener-
ated by the five coefficients δ, d, β, e, γ, that correspond to rows 5, 18, 9, 13, 8
of the above matrix of relations:

x3y = δt3, t5 = dx5, xzt = βy4, yz = ex3t, xyt = γz2 (5)

The distinguishing feature that makes this affine piece a reducible component
of A-Hilb (stuck at the origin) is that the monomial cone they generate is
not proper: here δ2dβeγ = 1, or in terms of rows of the matrix of relations:

2× (3 1 0 −3)
+ (−5 0 0 5)
+ (1 −4 1 1)
+ (−3 1 1 −1)
+ (1 1 −2 1)

= 0,

so these monomials span a whole vector space.

5.2 Lemma

Let Z be an A-cluster with monomial basis (4) parametrized by δ, d, β, e, γ of
(5). Then

yt2 = δdx2

xy2 = δ2dt
y3t = δ3d2x
y5 = δ5d3

x2z = δ2dβy2

zt2 = δ3d2βy
z2t = δ3d2βex3

z3 = δ7d4β2e

xt3 = δ3d2βγz
x6 = δ6d3β2γy
x5t = δ4d2β2γy3

x4t2 = δ7d4β2γ

Moreover,

δ = δ3dβeγ, that is, δ × (1− δ2dβeγ) = 0.

Near the monomial cluster (given by δ = d = β = e = γ = 0), necessarily
δ = 0. This is a copy of A4 in A-Hilb, consisting of clusters supported at the
origin. If δ2dβeγ = 1, the same relations define a cluster that can be free,
but that cannot approach the origin.

5.3 Proof

To clarify the statement: over any k algebra generated by δ, d, β, e, γ, any
A-cluster with this monomial basis has the stated relations.

This is similar to the simple example given on p. 17. For example, there
must be a relation yt2 = ξx2, and ξ is found by observing that x5 is basic, so
ξx5 = x3yt2 = δt5 = δdx5.

Finally, note that t4 is basic, so the final relation involving δ comes from

δt4 = x3yt = γx2z2 = δ2dβγy2z = δ3dβγt4

(the third equality is the relation x2z = δ2dβy2.)

d:=Random(KK); be:=Random(KK); e:=Random(KK); ga:=Random(KK);

Dimension(quo<RR|[x^3*y,t^5-d*x^5,x*z*t-be*y^4,y*z-e*x^3*t,

x*y*t-ga*z^2,y*t^2,x*y^2,y^3*t,y^5,x^2*z,z*t^2,z^2*t,z^3,x*t^3,

x^6,x^5*t,x^4*t^2]>);

de:=Random(KK); be:=Random(KK); e:=Random(KK); ga:=Random(KK);

d := (de^2*be*ga*e)^-1;

Dimension(quo<RR|[x^3*y-de*t^3, t^5-d*x^5, x*z*t-be*y^4,

y*z-e*x^3*t, x*y*t-ga*z^2, y*t^2-de*d*x^2, x*y^2-de^2*d*t,

y^3*t-de^3*d^2*x, y^5-de^5*d^3, x^2*z-de^2*d*be*y^2,

z*t^2-de^3*d^2*be*y, z^2*t-de^3*d^2*be*e*x^3,

z^3-de^7*d^4*be^2*e, x*t^3-de^3*d^2*be*ga*z,

x^6-de^6*d^3*be^2*ga*y, x^5*t-de^4*d^2*be^2*ga*y^3,

x^4*t^2-de^7*d^4*be^2*ga]>);

5.4 First reducible example
1
30

(1, 6, 10, 13) is the very interesting case with 158 affine pieces. No 16 is the
first reducible case. Note that it is reducible although the cone of invariant
ratios is a proper cone, defining the birational component. It has monomial
ideal generated by AS30[16]

x6, x4t2, x3y, x3z, x2z2, xyt, xt3, y2, yz, yt2, z3, z2t, zt2, t5

and complementary monomial basis MonomialBasis(quo<RR|AS30[16]>)

1, t, t2, t3, t4, z, zt, z2, y, yt,

x, xt, xt2, xz, xzt, xz2, xy, x2, x2t,

x2t2, x2z, x2zt, x2y, x3, x3t,

x3t2, x4, x4t, x5, x5t

I reorder the rows of its RelMatrix (by blundering about, in a process that
is still time-consuming and uncertain).

> M := RelMatrix(30,[1,6,10,13],AS30[16],Eig);

> M0 := Submatrix(M,[4,8,5,13,11,12,1,3,9,10,2,6,14,7,15],[1..4]);

> M0;

[ 3 1 0 -3]

[ 1 0 -1 3]

[ 3 0 1 -1]

[-3 0 2 1]

[-2 1 0 2]

[ 0 0 3 0]

[ 1 1 1 1]

[ 4 0 0 2]

[-2 2 -1 0] y^2 = rho*x^2*z

[-3 1 1 -1] y*z = la*x^3*t

[ 6 -1 0 0] x^6 = a*y

[ 2 0 2 -4] x^2*z^2 = l*t^4

[ 0 -1 1 2] z*t^2 = ka*y

[ 1 1 -2 1] x*y*t = ga*z^2

[-5 0 0 5] t^5 = d*x^5

> KernelMatrix(M0);

[ 1 0 0 0 0 0 0 0 0 0 0 -1 0 -1 0] x^3*y = l*ga*t^3 OK by t^4

[ 0 1 0 0 0 0 0 0 0 0 0 0 -1 -1 0] x*t^3 = ka*ga*z OK by z^2

[ 0 0 1 0 0 0 0 0 0 0 0 -1 -1 -1 0] x^3*z = l*ka*ga*t OK by z*t

[ 0 0 0 1 0 0 0 0 0 0 0 -1 0 0 -1] z^2*t = l*d*x^3 OK by x^5

[ 0 0 0 0 1 0 0 0 0 0 0 -1 0 -1 -1] y*t^2 = l*ga*d*x^2 OK by x^3

[ 0 0 0 0 0 1 0 0 0 0 0 -2 -1 -1 -1] z^3 = l^2*ka*ga*d OK by t

[ 0 0 0 0 0 0 1 0 0 0 0 -2 -1 -2 -1] x*y*z*t = l^2*ka*ga^2*d OK by 1

[ 0 0 0 0 0 0 0 1 0 0 0 -1 -2 -2 0] x^4*t^2 = l*ga*d*a OK by x^2

[ 0 0 0 0 0 0 0 0 1 0 0 -1 1 -1 -1]

[ 0 0 0 0 0 0 0 0 0 1 0 -1 1 0 -1]

[ 0 0 0 0 0 0 0 0 0 0 1 0 -2 -1 1]

The last 4 rows of M0 correspond to the relations

x2z2 = lt4, zt2 = κy, xyt = γz2, t5 = dx5.

The kernel matrix expresses the remaining 11 rows of M0 as integral linear
combinations of the last 4. For the first 8, these are positive,

ratios l, κ, γ, d base the invariant lattice M , and express as of the last 4,
so the ratios monomials in them.

y2 = ρ ∗ x2 ∗ z, y ∗ z = λ ∗ x3 ∗ t, x6 = a ∗ y.

The linear relations give the first 8 rows of M0 as last 4, with l, κ, γ, d as
close to a basis as one can come. The above statements

x3y = lγt3, . . . , x4t2 = lγda

are proved by the standard argument, involving cancelling the stated basic
monomial. On the other hand, the kernel matrix says that Row9 of M0

equals Row10 plus Row14, predicting the equality ρ = λγ on the birational
component, but this cannot be proved without further assumptions, and is
not true on the whole of A-Hilb. The reason this can’t be proved: there is
no deduction for the value of ρ, because every monomial except t kills x2z,
whereas

ρx2zt = y2t = so what?.

The way out is provided by the following lemma.

5.5 Lemma.

(I) 1. λκ = ld. 2. ad = κ2γ. 3. aλ = lκγ.
(II) (ρ − λγ) × various = 0. More specifically, 4. aρ = lκγ2 = aλγ. 5.

κρ = lγd = κλγ. 6. lρ = lλγ.

Corollary: either ρ = λγ, or a = κ = l = 0.

Proof.

1. z^2*t^2 = \la*\ka*x^3*t from

y*z = \la*x^3*t

z*t^2 = \ka*y

2. Similarly a*d = ka^2*ga; we can prove a*d*y = ka^2*ga*y from

x^6 = a*y

t^5 = d*x^5

a*d*y = d*x^6 = x*t^5 = ka*ga*z*t^2 = ka^2*g*y

3. Similarly for a*la = l*ka*ga; we can prove l*ka*ga*x^3*t = a*la*x^3*t

from

x^6 = a*y

y*z = la*x^3*t

a*la*x^3*t = a*y*z = x^6*z = l*ka*ga*x^3*t Q.E.D. (Lemma)

4. 5. 6.

rho*x^2*z = y^2

a*rho*x^2*z*t = a*y^2*t = x^6*y*t = ga*x^5*z^2 = l*ga*x^3*t^4 =

l*ka*ga^2*x^2*z*t,

(rho - la*ga)* various = 0.

proves a*rho = l*ka*ga^2 = a*la*ga

rho*x^2*z = y^2

ka*rho*x^2*z = ka*y^2 = y*z*t^2 = l*ga*d*x^2*z

proves ka*rho = l*ga*d = ka*la*ga

Take 7 variables $\rho,la,a,l,ka,ga,d$. Suppose that

la*ka=l*d, a*d=ka^2*ga, a*la=l*ka*ga

and EITHER $\rho = la*ga$ OR $a = ka = l = 0$. Then

$\rho,la,a,l,ka,ga,d$ parametrise A-clusters with

ideal

I:=[y^2-rho*x^2*z,y*z-la*x^3*t,x^6-a*y,x^2*z^2-l*t^4,

z*t^2-ka*y,x*y*t-ga*z^2,t^5-d*x^5,x^3*y-l*ga*t^3,

x*t^3-ka*ga*z,x^3*z-l*ka*ga*t,z^2*t-l*d*x^3,

y*t^2-l*ga*d*x^2,z^3-l^2*ka*ga*d,

x*y*z*t-l^2*ka*ga^2*d,x^4*t^2-l*ga*d*a];

Computer check:

rho:=Random(KK);

la:=Random(KK);

a:=Random(KK);

l:=Random(KK);

ka:=Random(KK);

ga:=Random(KK);

d:=Random(KK);

// uncomment one of these two lines

// a:=ka^2*ga*d^-1; la:=l*ka*ga*a^-1; rho := la*ga;

// a:=0; l:=0; ka:=0;

I:=[y^2-rho*x^2*z,y*z-la*x^3*t,x^6-a*y,x^2*z^2-l*t^4,

z*t^2-ka*y,x*y*t-ga*z^2,t^5-d*x^5,x^3*y-l*ga*t^3,

x*t^3-ka*ga*z,x^3*z-l*ka*ga*t,z^2*t-l*d*x^3,

y*t^2-l*ga*d*x^2,z^3-l^2*ka*ga*d,

x*y*z*t-l^2*ka*ga^2*d,x^4*t^2-l*ga*d*a];

Dimension(quo<RR|I>);

